

U. S. DEPARTMENT OF COMMERCE  
NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION  
NATIONAL WEATHER SERVICE  
NATIONAL METEOROLOGICAL CENTER

OFFICE NOTE 344

THE HYDROSTATIC CHECKING OF RADIOSONDE  
HEIGHTS AND TEMPERATURES

WILLIAM G. COLLINS  
DEVELOPMENT DIVISION

AND

LEV S. GANDIN  
UNIVERSITY OF MARYLAND

SEPTEMBER 1988

THIS IS AN UNREVIEWED MANUSCRIPT, PRIMARILY INTENDED FOR  
INFORMAL EXCHANGE OF INFORMATION AMONG NMC STAFF MEMBERS.

## Background

The present NMC operations include a hydrostatic check of radiosonde data, but it can at best result in the rejection by analysis codes of questionable data. It does this by assigning flags to the heights, used later by the analysis preprocessor in determining data quality. In a number of cases, the operational code results in bad data being kept, good data being rejected, or good data being rejected in addition to bad data in a report. The approach reported in this paper is designed to identify hydrostatic errors that can be corrected, and to make the correction. It is recommended that it be incorporated into operations just prior to the operational 'hydrostatic check', HYDROCHEK. In this way, the corrected data will not get flagged, and will be used in the subsequent analysis. This procedure will not correct any of the present failings of HYDROCHK, which must await its replacement.

A total of about 70 reports are found to have hydrostatic errors for each data cycle. Of these, about 25 have errors that can be confidently corrected, being roughly evenly divided between height and temperature corrections. These data are particularly important as they can be located throughout the globe. Table 1 for a May 1988 test shows that a majority of the hydrostatic errors were from stations in the U.S.S.R., India, and China. Table 2 shows the distribution by station block for 15 cases in May, June, and July 1988. The same areas are seen to predominate.

Table 1. Hydrostatic Errors by Region

<u>Region</u>	<u>Average No.</u>	<u>Percent</u>
U.S.S.R.	8.0	20
India	8.3	16
China	7.9	15
Indonesia	2.8	5
Australia	2.0	4

---

In the complex quality control being designed at NMC (Gandin, 1988), the hydrostatic check forms one of several quality control steps to be performed in parallel, before decisions are made regarding data rejection or correction. The hydrostatic check is unique in being such a powerful check because of the redundancy of height and temperature data. It is because of its power that the hydrostatic check by itself can lead to corrections of the data. The next section of the paper will show the underlying theory used to determine hydrostatic errors and to make corrections. The next section will show some examples of errors and define

the methods for their correction. The following sections will give statistical summaries of the errors encountered in tests, and outline implementation plans.

Table 2. Number of Hydrostatic Errors by WMO Block Number for 15 Cases in May, June and July 1988.

units-->		0	1	2	3	4	5	6	7	8	9
tens-->	00	-	2	3	0	1	-	0	3	4	0
	10	2	0	1	1	-	15	7	13	-	-
	20	0	2	3	5	9	9	7	9	9	12
	30	9	13	3	2	3	6	7	13	25	-
	40	5	19	48	37	21	0	4	19	17	-
	50	4	14	13	7	14	6	21	15	6	12
	60	15	5	4	5	0	0	0	3	3	-
	70	1	7	3	0	0	-	11	-	4	-
	80	2	1	5	5	1	2	0	3	0	3
	90	-	19	-	4	24	-	13	7	0	-

### Theory

The hydrostatic check is based on the redundancy of the reported heights and temperatures. The hydrostatic equation is used at station locations to determine the heights from the temperatures observed at known pressures. If the values received at NMC do not agree hydrostatically, then an error has occurred either during the computation at the observation location, during entry into the communications system, during the transfer of the data to NMC, or in our decoding of the data. No matter what the cause of the error, it is in many cases possible to determine that an error has occurred, and to find a correction. In many cases, it is also possible to suggest the nature of the error.

The hydrostatic error is determined by the left and right hand sides of the hydrostatic equation not agreeing. The disagreement is called the hydrostatic residual and is defined by

$$s_i^{i+1} = z_{i+1} - z_i - A_i^{i+1} - B_i^{i+1} (T_i + T_{i+1}).$$

with

$$A_i^{i+1} = (RT_0/g) \ln(p_i / p_{i+1})$$

$$B_{i,i+1} = (R/2g) \ln(p_i / p_{i+1})$$

To = 273.16 K, R is the dry gas constant, and g is the acceleration of gravity. Temperatures are measured in degrees Celsius, and the difference between temperature and virtual temperature is accounted for by not requiring exact hydrostatic agreement.

The hydrostatic check described in this note only uses mandatory level data in the check, and assumes a linear in logarithm of pressure variation of temperature between mandatory levels. Furthermore, the temperatures are considered to be dry. Therefore, we do not expect the residuals to be zero, but rather they should be small, with maximum allowable values to be determined experimentally.

The possibility of error correction comes from comparing a consecutive pair of hydrostatic residuals. If the sum of the residual pair is small, but the residuals are large, then the intermediate height has an error. Conversely, if the difference of the residuals, divided by the layer B's, is small, then there is an error in the intermediate temperature. These ideas will be made more precise in the next section.

### Application

Confident height or temperature corrections are made when residuals are large and the adjusted values would lead to hydrostatic residuals that are small. This statement is made precise by the following discussion. First, define the following quantities:

$$s_z = |s_{i,i+1} + s_{j,j+1}|$$

$$S_z = (B_{i,i+1} + B_{j,j+1}) * DTALL$$

$$s_T = |s_{i,i+1} / B_{i,i+1} - s_{j,j+1} / B_{j,j+1}|$$

$$S_T = (1/B_{i,i+1} + 1/B_{j,j+1}) * DZALL$$

where  $i$  to  $i+1$  and  $j$  to  $j+1$  are indices for the layers below and above the data level in question, the  $B$ 's are summed over the whole layers  $i$  to  $i+1$  and  $j$  to  $j+1$  when there are intermediate missing data, and  $DTALL$  and  $DZALL$  are the allowable temperature and height errors: 5 Kelvin and 15 meters. We may call  $S_z$  the admissible difference between the height corrections for the layers above and below as suggested by the residuals, and  $S_T$  the admissible difference in temperature correction for the layers below and above.

### Confident height correction

No corrections are made unless a hydrostatic residual exceeds  $SBIG$ , an experimentally determined value for the layer in question, which is presently about 7 times the standard deviation of residual. The values are shown in Table 3.

A confident height correction is made when

$$s_z < S_z$$

and the value of the smaller of the  $s$ 's is greater than  $0.5*SBIG_j$ . For the mandatory layers,  $SBIG$  varies from 30 meters (1000-850 mb) to 100 meters (20-10 mb). If a value is needed for the sum of more than one layer (as it is when height or temperature values are missing in a profile), the value is obtained as follows:

$$SBIG(\text{layers } n \text{ to } m) = \text{SQRT} \left[ \sum_{i=n}^m SBIG_i^{i+1 \ 2} \right]$$

The height correction that is made is

$$ZCOR_J = -.5 * (s_i^{i+1} - s_j^{j+1})$$

where  $i$  to  $i+1$  are the indices for the higher pressure layer,  $j$  to  $j+1$  are the indices for the lower pressure layer, and  $J$  is the index for the corrected height. The values of these and all suggested height corrections are rounded to the nearest meter at 1000, 850 and 700 mb, and rounded to the nearest 10 meters above, where no correction is made unless the magnitude is at least 100 meters.

Table 3. Admissible hydrostatic residuals for mandatory layers.

<u>PRESSURE</u>	<u>ADMISSIBLE RESIDUAL</u>
1000	
	65.
850	
	35.
700	
	50.
500	
	35.
400	
	40.
300	
	35.
250	
	40.
200	
	50.
150	
	85.
100	
	70.
70	
	70.
50	
	80.
30	
	70.
20	
	100.
10	

Table 4 shows an example of a confident height correction. The parameters are pressure, height, corrected height, temperature, corrected temperature, layer residual, height correction, temperature correction, and "correction type". In most of the similar tables, only a portion of the reported profile will be shown. The pair of residuals clearly indicate a height correction of 3000. meters. The error most likely results from a simple key typing mistake. The next example, shown in Table 5, is a confident height correction in which digits were transposed.

Table 4. Example of confident height correction in which a key typing mistake was made.

IDENT	HEIGHT	NEW-HEIGHT	TEMP.	NEW-TEMP	RESID	ZCOR	TCOR	TYP
50	21090.	21090.	-49.3	-49.3		0.0	0.0	
					-3007.4			
30	21420.	24420.	-50.9	-50.9		3000.	0.0	1
					2999.1			
20	27080.	27080.	-47.3	-47.3		0.0	0.0	

Table 5. Example of confident height correction in which digits were interchanged.

IDENT	HEIGHT	NEW-HEIGHT	TEMP.	NEW-TEMP	RESID	ZCOR	TCOR	TYP
300	9770.	9770.	-24.9	-24.9		0.0	0.0	
					-902.8			
250	10170.	11070.	-33.5	-33.5		900.	0.0	1
					900.9			
200	12600.	12600.	-44.9	-44.9		0.0	0.0	

Confident temperature correction

A confident temperature correction is made when

$$\frac{s}{T} < \frac{S}{T}$$

and the value of the smaller  $s$  exceeds  $0.5 \cdot \text{SBIG}$ . The temperature correction that is made is

$$\text{TCOR}_J = .5 \left( s_{i+1} / \text{BSUM}_i + s_{j+1} / \text{BSUM}_j \right)$$

where  $i$  to  $i+1$  are the indices for the higher pressure layer,  $j$  to  $j+1$  are the indices for the lower pressure layer, and  $J$  is the index for the corrected height. The values of all suggested temperature corrections are rounded to the nearest 0.1 degree. Table 6 shows an example of a confident temperature correction. The correct value is most likely 21.6, and the '2' was mistakenly entered as an '8'.

The examples have focused attention on the portion of profiles where errors appeared. Cases occur not infrequently where there is more than one confident height and/or temperature correction, and the procedures outlined in this note treat these situations just as well as isolated

errors, so long as one level of good data intervenes. There will be more discussion later of complicated cases.

Table 6. Example of a confident temperature correction.

IDENT	HEIGHT	NEW-HT	TEMP.	NEW-T	RESID	ZCOR	TCOR	TYP
1000	69.	69.	28.0	28.0		0.0	0.0	
					-144.1			
850	1486.	1486.	81.6	21.4		0.	-60.2	2
					-170.1			
700	3132.	3132.	10.8	10.8		0.0	0.0	

In some cases it is possible to make a sign correction in temperature. If  $|2 \cdot T + TCOR| < DTALL$ , then the sign is in error and must be corrected. Table 7 shows an example of a temperature sign correction. Many temperature corrections are sign corrections. A check is also made to make sure that a corrected temperature does not produce a statically unstable layer. This is accomplished by checking the lapse rates above and below a corrected level.

Table 7. Example of a confident temperature correction with sign error.

IDENT	HEIGHT	NEW-HT	TEMP.	NEW-T	RESID	ZCOR	TCOR	TYP
700	3081.	3081.	1.2	1.2		0.0	0.0	
					-144.8			
500	5710.	5710.	15.4	-15.4		0.0	-29.8	2
					-98.4			
400	7360.	7360.	-26.7	-26.7		0.0	0.0	

Other corrections

In cases where a confident correction cannot be made, suggested corrections can sometimes be given. This information will be used along with the horizontal and vertical checks by the Decision Making Algorithm to determine further confident corrections.

If the error is in the bottom layer, it is most likely that the lowest temperature or height is in error, but usually not both. Accordingly, if the residual in the lowest layer is greater than SBIG, and is less than .5\*SBIG in the second layer, the amount that the independent temperature or height must be changed to make the residual



equal to 0. is given as the suggested change. The suggested corrections are

$$ZCOR = s \frac{i+1}{J \quad i}$$

$$TCOR = s \frac{i+1}{J \quad i} / BSUM \frac{i+1}{i}$$

with the temperature and height rounded as described before. There is the additional strong possibility that a computational error giving incorrect thickness in the lowest layer is reflected in all heights above having the same error, rather than the lowest height being wrong. However, a choice between the possibilities can only be determined with the help of other Complex Quality Control components, particularly the horizontal check of height. Tables 8 and 9 give examples of errors in the bottom layer. Examination shows that the error is in the height (or possibly thickness) in the first, and in the sign of temperature in the second.

Table 8. Example of hydrostatic error in the bottom layer, due to height error.

IDENT	DATE	NEW-HT	TEMP.	NEW-T	RESID	ZCOR	TCOR	TYP
20069	12Z 31 AUG 1988	1000	112.	112.	-0.1	-85.0	-355.6	4
					-84.8			
850	1318.	1318.	-3.9	-3.9		0.0	0.0	

Table 9. Example of hydrostatic error in the bottom layer, due to temperature error.

IDENT	DATE	NEW-HT	TEMP.	NEW-T	RESID	ZCOR	TCOR	TYP
29231	12Z 31 AUG 1988	1000	176.	176.	-22.9	110.0	46.3	4
					110.2			
850	1570.	1570.	16.0	16.0		0.0	0.0	

Independent height and temperature corrections are also suggested for a large hydrostatic residual in the top layer when the next lower residual is less than .5\*SBIG. The suggested corrections are

$$ZCOR = -s \frac{i+1}{J \quad i}$$

$$TCOR = \frac{s_{i+1}}{J} - \frac{s_{i+1}}{BSUM_i}$$

Examples of height and temperature errors at the top are given in Tables 10 and 11.

Table 10. Example of hydrostatic error in the top layer, due to height error.

```
IDENT = 71816  DATE = 12Z 31 AUG 1988
PRESS HEIGHT NEW-HT TEMP. NEW-T  RESID  ZCOR  TCOR  TYP
  20 26910. 26910. -48.7 -48.7          0.0  0.0
          310.1
  10 31840. 31840. -42.5 -42.5        -310.0  30.5  5
-----
```

Table 11. Example of hydrostatic error in the top layer, due to temperature error.

```
IDENT = 38879  DATE = 12Z 30 AUG 1988
PRESS HEIGHT NEW-HT TEMP. NEW-T  RESID  ZCOR  TCOR  TYP
  300 9470. 9470. -42.5 -42.5          0.0  0.0
          -127.0
  250 10700. 10700.  4.4  4.4          130.0 -47.6  5
-----
```

A pair of corrections can be suggested for an intermediate layer in which the conditions for a confident correction are not met, i.e. where neither

$$s < S$$

$$z < Z$$

nor

$$s < S$$

$$T < T$$

The suggested corrections in this case are those which, when taken together, make the pair of residuals equal 0. It is rare to encounter a case where careful examination shows that simultaneous height and temperature errors have occurred at the same level, but the pair of suggested corrections can serve as a guide in further tests. These corrections are suggested only under the further restriction

that the smaller of the residuals is greater than .5\*SBIG.  
They are

$$ZCOR = \frac{(BSUM_{i+1} * s_j - BSUM_{j+1} * s_i)}{BSUM_{i+1} + BSUM_{j+1}}$$

and

$$TCOR = \frac{s_{i+1} + s_{j+1}}{BSUM_{i+1} + BSUM_{j+1}}$$

An example of a suggested correction pair is shown in Table 12. In this case, both corrections of the pair must be applied or the layer above will be statically unstable.

Table 12. Example of hydrostatic error which does not satisfy the criteria for being confidently a height or temperature error.

```
IDENT = 98646  DATE = 00Z 14 SEP 1988
PRESS HEIGHT NEW-HT TEMP. NEW-T  RESID  ZCOR  TCOR  TYP
  700  3156.  3156.  10.0  10.0          0.0  0.0
          72.0
  500  5990.  5990.   4.2   4.2        -100.0  -5.7   3
          -119.0
  400  7620.  7620. -15.3 -15.3          0.0  0.0
-----
```

In addition to these suggested corrections, the presence of isolated large residuals is detected, but in this case, it is rarely possible to suggest a correction without the help of other CQC elements. An example of an isolated large residual is shown in Table 13. In this case it is likely that the error is in the computation of the 100 to 70 mb thickness.

In some cases of more complicated error it is possible to automatically make corrections. The following case will serve as an example. Methods to handle such a case have been developed, but not yet included in the hydrostatic check code. The example, shown in Table 14, has two places in the profile where there are three consecutive large residuals. The suggested corrections are those found by the

new procedure, not by the present code, which does not suggest confident corrections.

Table 13. Example of large isolated residual.

IDENT = 37789		DATE = 12Z 30 AUG 1988						
PRESS	HEIGHT	NEW-HT	TEMP.	NEW-T	RESID	ZCOR	TCOR	TYP
150	14000.	14000.	-51.7	-51.7		0.0	0.0	
					-9.8			
100	16600.	16600.	-55.1	-55.1		0.0	0.0	
					-429.7			
70	18440.	18440.	-56.7	-56.7		0.0	0.0	6
					-14.1			
50	20570.	20570.	-54.5	-54.5		0.0	0.0	

Table 14. Example of complicated corrections.

IDENT = 87623		DATE = 12Z 1 SEP 1988						
PRESS	HEIGHT	NEW-HT	TEMP.	NEW-T	RESID	ZCOR	TCOR	TYP
1000	259.	259.	-0.3	-0.3		0.0	0.0	
					392.4			
850	1155.	1547.	-4.7	-4.7		392.0	0.0	
					640.3			
700	3306.	3079.	-10.3	-10.3		-227.0	0.0	
					-227.4			
500	5610.	5610.	-22.3	-22.3		0.0	0.0	
					3.1			
400	7220.	7220.	-32.3	-32.3		0.0	0.0	
					2.9			
300	9190.	9190.	-47.1	-47.1		0.0	0.0	
					2.3			
250	10400.	10400.	-46.9	-46.9		0.0	0.0	
					-7.7			
200	11860.	11860.	-50.3	-50.3		0.0	0.0	
					88.3			
150	13720.	13720.	-75.5	-54.5		0.0	21.0	
					-5157.6			
100	11000.	16260.	-60.3	-60.3		5260.0	0.0	
					5264.2			
70	18490.	18490.	-59.9	-59.9		0.0	0.0	
					3.1			
50	20580.	20580.	-62.9	-62.9		0.0	0.0	
					30.1			
30	23750.	23750.	-63.7	-63.7		0.0	0.0	
					18.9			
20	26260.	26260.	-63.1	-63.1		0.0	0.0	

## Selection of Residual Pair

All of the hydrostatic error detection techniques use a residual pair for their detection. When there is more than one error in a profile, it is crucial that the proper order of residual pair consideration be chosen since the order can affect the conclusions of error type. In some cases it is necessary to consider three consecutive residuals for proper error determination, but the present method does not include that further complication. This section will discuss the priority system used to select pairs of residuals for error determination.

Some general principles are used in residual selection. The selection of a residual for consideration is based upon the size of the residual relative to the admissible

residual,  $s_{i+1} / \text{SBIG}_i$ . The SBIG's themselves are related

to the standard deviation of  $s_{i+1}$ . Therefore, we are

interested in the magnitude of the residual, relative to its climatological variation at the level in question. We give priority to residuals having neighbors that have not already been tested (mid), since then the powerful methods outlined in the previous section may be used.

During our examination of radiosonde data it was found that there are many reports that have not been successfully decoded at all levels, leading to missing data at upper levels. Since the reports arrive in parts (1000-100 mb and 70 mb up), it is often the case that an upper part can be decoded successfully, while levels of the lower part have been lost. This gives the appearance of a report with several layers missing, usually ending with 100 mb. When the hydrostatic check is applied to the levels bounding such a data gap, and particularly if the tropopause is within this layer, the check is not of much value. Therefore, priority is also given to layers without missing data (thin), since then the assumption of linearity of the temperature profile is more accurate and the conclusions of the analysis are more definite.

The specifics of the selection priority may be outlined as follows:

Selection of large residual--

It must not already have been tested.

Top priority--greatest  $s/\text{SBIG}$ , thin, mid.

2nd priority--greatest  $s/\text{SBIG}$ , top or bottom.

3rd priority--greatest  $s/\text{SBIG}$ .

Selection of neighbor--

- It must not already have been tested.
- If only one neighbor is untested, pick it.
- If only one neighbor is thin, pick it.
- Pick neighbor with largest  $s/SBIG$ .

If there are no untested neighbors or no untested residuals, the checking ends.

The FORTRAN codes for important subroutines for the hydrostatic corrections are given in the Appendix at the end of this Note.

### Statistical results

This section will summarize some of the results of the new hydrostatic check. Out of an average of roughly 600 reports, about 70 were found to contain one or more hydrostatic errors. The following Tables 15 and 16 show the distribution of errors by error type and by pressure. It is noted that 37% of the errors can be confidently corrected.

In all the test cases, there were only a few decisions for a confident correction which close examination showed to be in error. They were a result of problems with the sounding that are probably beyond the capability of any procedure to rehabilitate. Therefore, there were no instances when a correction by the new hydrostatic check, and its data correction, introduced additional errors to the data. On the other hand, the overwhelming majority of its corrections to the data passed the subsequent checking by HYDROCHK (see DiMego et al, 1985), and therefore an increase of good data to the NMC operational codes would have resulted had they been run following these procedures.

Table 15. Distribution of Hydrostatic Errors by Type

No.	Error Type	Average No.	Percent
1	confident height correction	11.3	16
2	confident temperature correction	14.9	21
3	not confident, T, z pair of corrections given	6.0	8
4	error at bottom	8.0	11
5	error at top	13.6	19
6	large isolated residual	8.6	12
11	nearly confident height corr.	6.0	8
12	temperature correction diagnosed to give unstable lapse rate	3.0	5

-----

Table 16. Distribution of Hydrostatic Errors by Pressure

Pressure	Average No.	Percent
1000	2.2	3
850	5.0	8
700	3.0	5
500	3.8	6
400	5.2	8
300	5.4	8
250	4.2	7
200	2.4	4
150	5.8	9
100	2.8	4
70	9.2	15
50	3.8	6
30	2.4	4
20	3.0	5
10	5.0	8

The present 'hydrostatic check' code, HYDROCHK, does more than a hydrostatic check. It also checks for layers with lapse rates outside specified ranges, it checks values against climatological ranges, and it checks wind shears. When it finds a problem, it flags the datum. In some cases, it used significant level data to try to reconstruct part of a profile of temperatures, and to modify and check mandatory level temperatures. Again, flags are used to indicate its decisions. Eight cases in May 1988 were used to investigate the validity of HYDROCHK's decisions. The following Table 17 summarizes those results.

Table 17. Reaction of the Present System to the Flags Set by HYDROCHK

Reaction of present system	Average No.	Percent
purged the datum (the proper response)	7.4	52
purged the level (z and T)	3.5	20
purged other data instead	1.3	7
purged other data in addition	2.0	11
purged many data	2.4	14
no reaction	.9	5
attempted to correct	.1	1

The fact that the HYDROCHK almost always has some reaction to hydrostatic errors is significant since it is proposed to include the new hydrostatic check just prior to HYDROCHK. As mentioned above, the tests show that the

changes made by the new code are invariably seen to be acceptable to HYDROCHK.

There are many possible sources leading to errors in datum received at NMC, but the hydrostatic check can only detect those that lead to an inconsistency between the height and temperature fields. The possible reasons include the computation of the heights from the temperatures, transcription to the communication lines, and communication errors. Refer to Table 1, which shows the distribution of hydrostatic errors by geographical region for the 8 cases in May 1988. It is seen that the largest number of errors come from regions with the least automation. Generally, there are no errors from the U.S., so that the improvement to the analyses that will be effected by the introduction of the new hydrostatic check will be in other areas. Its impact on forecast quality will most likely usually be small, but not necessarily so.

#### Summary

This note has described a hydrostatic check which can be used to correct certain types of hydrostatic errors. Tests have shown the check to be reliable, and further testing in the operational framework are under way. It is proposed to perform the HSC just prior to the operational code HYDROCHK. In this way, the data that are corrected will not be flagged as erroneous by HYDROCHK, and they will likely be used in NMC's analyses and forecasts.

During a month-long test, the confident height and temperature corrections that the hydrostatic check would make to the data will be scrutinized by monitoring analysts of the Meteorological Operations Division. Following a positive result from that test, the HSC will be implemented. This procedure is seen as an interim measure for the following reasons. First of all, it is desirable that all the functions of HYDROCHK be taken over by the new code. But also, the HSC is designed to be one component of the Complex Quality Control. And in that framework, it may not be appropriate to make changes to the data without the results of the vertical and horizontal checks. The approach taken here, of modifying the data as a result of a single test, is only possible because of the power of the hydrostatic check. However, even it is not infallible, and in the future, when more components of the Complex Quality Control are available, and the Decision Making Algorithm has been written, we may reverse the present decision, and not allow the HSC alone to make data corrections.

Future work will concentrate first on the horizontal and then the vertical check of radiosonde data, followed by the Decision Making Algorithm for these data. The reader is



referred to Gandin (1988) for further discussion. Each data type has its special types of errors and methods of cross-checking with other information, so that the development must be independent, but parallel.

## REFERENCES

Gandin, L.S., 1988: Complex quality control of meteorological observations. Mon. Wea.Rev., 116(5),1137-1156.

DiMego, G.J., P.A. Phoebus, and J.E. McDonell, Data processing, and quality control for optimum interpolation analyses at the National Meteorological Center, 1985: NOAA, National Weather Service, National Meteorological Center, Office Note 306.

## APPENDIX

Code for subroutines CORECT, FNEXT, TEST, and RESID begins on the following page.

SUBROUTINE CORECT(Z, T, ZCOR, TCOR, ZC, TC, S, SBIG,  
& LEV, ICTYP, IETYP)

C\$\$\$ SUBPROGRAM DOCUMENTATION BLOCK

C  
C SUBPROGRAM: CORECT HYDROSTATIC CORRECTION TO DATA  
C PRGMMR: W. COLLINS ORG: W/NMC22 DATE: 88-09-14  
C

C ABSTRACT: THIS SUBROUTINE MAKES CORRECTIONS TO HEIGHTS AND  
C TEMPERATURES IN A RADIOSONDE REPORT BASED UPON  
C A HYDROSTATIC CHECK.  
C

C PROGRAM HISTORY LOG:  
C 88-09-14 ORIGINAL W. COLLINS  
C

C USAGE: CALL CORECT(Z, T, ZCOR, TCOR, ZC, TC, S, SBIG,  
C LEV, ICTYP, IETYP)

C INPUT ARGUMENT LIST:  
C Z - HEIGHT PROFILE  
C T - TEMPERATURE PROFILE  
C SBIG - ADMISSIBLE RESIDUAL FOR LAYER (M)  
C LEV - NUMBER OF LEVELS TO CONSIDER  
C

C OUTPUT ARGUMENT LIST:  
C ZC - CORRECTED HEIGHT PROFILE (METERS)  
C TC - CORRECTED TEMPERATURE PROFILE (CELSIUS)  
C S - NEW HYDROSTATIC RESIDUAL FOR LAYER (M)  
C ICTYP - CORRECTION TYPE CODE  
C 0 = NO CORRECTION  
C 1 = CONFIDENT HEIGHT CORRECTION  
C 2 = CONFIDENT TEMPERATURE CORRECTION  
C 3 = Z, T CORRECTIONS TO MAKE RESIDS = 0.  
C 4 = BOTTOM LAYER CORRECTION CHOICE  
C 5 = TOP LAYER CORRECTION CHOICE  
C 6 = ISOLATED LARGE RESIDUAL  
C 11 = HEIGHT CORRECTION .LT. 100 M  
C 12 = TEMPERATURE CORRECTION, GIVING INSTABILITY  
C IETYP - ERROR TYPE CODE (NOT YET DEFINED)  
C

C ATTRIBUTES:  
C LANGUAGE: VS FORTRAN  
C MACHINE: NAS  
C

C\$\$\$

DIMENSION Z(1), T(1), ZC(1), TC(1), ICTYP(1), IETYP(1)  
DIMENSION ZCOR(1), TCOR(1), S(1), SBIG(1),  
& BSUM(54), ISL(54), ISU(54), ITST(54)  
COMMON /CONSTS/ R, G, TO, A(54), B(54), SS(54)  
COMMON /LEVEL/ MBOGUS, NPLVL, IPLVL(55)  
DATA ZMSG /99999./, TMSG /9999.9/, ZMAX /90000./,  
& TMAX /9000./, DZALL /15./, DTALL /5./,  
& CP /1004.5/, PO /1000./, THLIM /-.05/  
LEVM = LEV - 1  
ROCP = R/CP

C  
C FIND THE HIGHEST NON-MISSING LEVEL. SET INDEX = KMAX.  
C

DO 10 K=1, NPLVL  
KK = NPLVL - K + 1  
KMAX = KK  
IF((Z(KK).LT.ZMAX).AND.(T(KK).LT.TMAX)) GO TO 20

```

10 CONTINUE
C
C   COPY PROFILES INTO OUTPUT PROFILES.
C
20 CONTINUE
   DO 30 K=1,LEV
       ZC(K) = Z(K)
       TC(K) = T(K)
30 CONTINUE
C
C   ZERO SOME QUANTITIES.
C
   DO 40 K=1,LEVM
       ICTYP(K) = 0
       IETYP(K) = 0
40 CONTINUE
C
C   ZERO CORRECTIONS, INDEX LIST, RESIDUALS.
C
   DO 50 K=1,NPLVL
       ZCOR(K) = 0.
       TCOR(K) = 0.
       ITST(K) = 0
       S(K) = 0.
50 CONTINUE
C
C   COMPUTE RESIDUALS
C
100 CONTINUE
   CALL RESID(ZC,TC,KMAX,S,SBIG,BSUM,ISL,ISU,KRES)
C
C   FIND THE (NEXT) LARGEST TEMPERATURE RESIDUAL, S(II)/SBIG(II)
C   AND LARGEST ADJACENT TEMPERATURE RESIDUAL, S(JJ)/SBIG(JJ)
C   AMONG LAYERS THAT HAVE NOT ALREADY BEEN TESTED.
C
300 CONTINUE
   CALL FNEXT(S,SBIG,ITST,KRES,ISL,ISU,II,JJ,SI,SJ)
C
C   SEE IF LARGEST RESIDUAL IS LARGE.
C
   IF(ABS(SI).LE.SBIG(II)) RETURN
C
C   CHECK TO SEE IF ALL LEVELS HAVE BEEN CHECKED.
C
   CALL TEST(ITST,KRES,IT)
   IF((II.EQ.0).OR.(IT.EQ.KRES)) RETURN
C
C   TEST FOR ERROR TYPE. FIRST COMPUTE SOME QUANTITIES.
C
   IF(JJ.NE.0) THEN
       IF(II.LT.JJ) THEN
           JMM = ISL(II)
           JM = ISU(II)
           JP = ISU(JJ)
           SIGN = 1.0
       ELSE
           JMM = ISL(JJ)
           JM = ISU(JJ)
           JP = ISU(II)
           SIGN = -1.0

```

```
ENDIF
SHGT = ABS(SI + SJ)
SSHGT = (BSUM(II)+BSUM(JJ)) * DTALL
STMP = ABS(SI/BSUM(II) - SJ/BSUM(JJ))
SSTMP = (1./BSUM(II)+1./BSUM(JJ)) * DZALL
ENDIF
JT = ISU(II)
JB = ISL(II)
ICT = 0
```

```
C
C      CONFIDENT HEIGHT CORRECTION (ICTYP = 1).
C
```

```
IF((ABS(SJ).GT.0.5*SBIG(JJ)).AND.JJ.NE.0.
& AND.(SHGT.LT.SSHGT)) THEN
ZCOR(JM) = -0.5 * SIGN * (SI - SJ)
ROUND TO NEAREST 10 METERS ABOVE 700 MB
ROUND TO NEAREST METER BELOW 500 MB
IF(JM.LE.3) THEN
  ZCOR(JM) = ANINT(ZCOR(JM))
  ICTYP(JM) = 1
  ICT = 1
ELSE
  ZCOR(JM) = 10. * ANINT(ZCOR(JM)/10.)
  IF(ABS(ZCOR(JM)).GE.100.) THEN
    ICTYP(JM) = 1
    ICT = 1
  ELSE
    ZCOR(JM) = 0.
    ICTYP(JM) = 11
    ICT = 11
  ENDIF
ENDIF
ZC(JM) = ZC(JM) + ZCOR(JM)
ITST(II) = 1
ITST(JJ) = 1
```

```
C
C      CONFIDENT TEMPERATURE CORRECTION (ICTYP = 2).
C
```

```
ELSE IF((ABS(SJ).GT.0.5*SBIG(JJ)).AND.JJ.NE.0.
& AND.(STMP.LT.SSTMP)) THEN
TCOR(JM) = 0.5 * (SI/BSUM(II)+SJ/BSUM(JJ))
TCOLD = TC(JM)
IF(ABS(2.*TC(JM)+TCOR(JM)).LT.DTALL) THEN
  TC(JM) = -TC(JM)
ELSE
  ROUND TO NEAREST 1/10 DEGREE
  TCOR(JM) = 0.1 * ANINT(10.*TCOR(JM))
  TC(JM) = TC(JM) + TCOR(JM)
ENDIF
```

```
C
C      MAKE SURE THAT THE LAPSE RATES ABOVE AND BELOW
C      ARE ADIABATICALLY STABLE.  IF NOT, RESTORE
C      THE ORIGINAL TEMPERATURE, TCOLD.
```

```
C
C      ALSO, CHECK FOR TEMPERATURE LAPSES ABOVE AND
C      BELOW OF SAME SIGN.
```

```
C
C      ALSO, DO NOT GIVE CORRECTIONS FOR LAYERS WITH
C      MISSINGS.
C
```

```

IF(JP.NE.0.AND.JMM.NE.0) THEN
  PJP = IPLVL(JP)
  THJP = (T0+TC(JP)) * (P0/PJP)**ROCP
  PJM = IPLVL(JM)
  THJM = (T0+TC(JM)) * (P0/PJM)**ROCP
  THOLD = (T0+TCOLD) * (P0/PJM)**ROCP
  PJMM = IPLVL(JMM)
  THJMM = (T0+TC(JMM)) * (P0/PJMM)**ROCP
  IF(THJP-THJM.GT.THLIM
&   .AND.THJM-THJMM.GT.THLIM
&   .AND.JP-JMM.EQ.2
&   .AND.((TC(JP)-TC(JM))*(TC(JM)-TC(JMM))).GT.0.)
&   THEN
    ICTYP(JM) = 2
    ICT = 2
    ITST(II) = 1
    ITST(JJ) = 1
  ELSE
    TC(JM) = TCOLD
    ICTYP(JM) = 12
    ICT = 12
    ITST(II) = 1
    ITST(JJ) = 1
  ENDIF
ENDIF
ENDIF

```

```

C
C A CONFIDENT CORRECTION CANNOT BE MADE, THEREFORE:
C COMPUTE CORRECTION PAIR MAKING S-PAIR = 0. (ICTYP = 3)
C

```

```

ELSE IF(ABS(SJ).GT.0.5*SBIG(JJ)
& .AND.JJ.NE.0) THEN
  ZCOR(JM) = SIGN * (BSUM(II)*SJ-BSUM(JJ)*SI)
&   /((BSUM(II) + BSUM(JJ))
C ROUND TO NEAREST 10 METERS ABOVE 700 MB
C ROUND TO NEAREST METER BELOW 500 MB
  IF(JM.LE.3) ZCOR(JM) = ANINT(ZCOR(JM))
  IF(JM.GT.3) ZCOR(JM) = 10. * ANINT(ZCOR(JM)/10.)
  TCOR(JM) = (SI + SJ) / (BSUM(II) + BSUM(JJ))
C ROUND TO NEAREST 1/10 DEGREE
  TCOR(JM) = 0.1 * ANINT(10.*TCOR(JM))
  ICTYP(JM) = 3
  ICT = 3
  ITST(II) = 1
  ITST(JJ) = 1

```

```

C
C CORRECTION AT BOTTOM. (ICTYP = 4)
C

```

```

ELSE IF((II.EQ.1).
& AND.((JJ.EQ.0).OR.((JJ.NE.0).AND.(ABS(SJ).
& LT.0.5*SBIG(JJ)))) THEN
  ZCOR(JB) = SI
C ROUND TO NEAREST 10 METERS ABOVE 700 MB
C ROUND TO NEAREST METER BELOW 500 MB
  IF(JB.LE.3) ZCOR(JB) = ANINT(ZCOR(JB))
  IF(JB.GT.3) ZCOR(JB) = 10. * ANINT(ZCOR(JB)/10.)
  TCOR(JB) = SI/BSUM(II)
C ROUND TO NEAREST 1/10 DEGREE
  TCOR(JB) = 0.1 * ANINT(10.*TCOR(JB))
  ICTYP(JB) = 4
  ICT = 4

```

```

      ITST(II) = 1
C
C CORRECTION AT TOP. (ICTYP = 5)
C
      ELSE IF((II.EQ.KRES).
& AND.((JJ.EQ.0).OR.((JJ.NE.0).AND.(ABS(SJ).
& LT.0.5*SBIG(JJ)))) THEN
      ZCOR(JT) = -SI
C      ROUND TO NEAREST 10 METERS ABOVE 700 MB
C      ROUND TO NEAREST METER BELOW 500 MB
      IF(JT.LE.3) ZCOR(JT) = ANINT(ZCOR(JT))
      IF(JT.GT.3) ZCOR(JT) = 10. * ANINT(ZCOR(JT)/10.)
      TCOR(JT) = SI/BSUM(II)
C      ROUND TO NEAREST 1/10 DEGREE
      TCOR(JT) = 0.1 * ANINT(10.*TCOR(JT))
      ICTYP(JT) = 5
      ICT = 5
      ITST(II) = 1
C
C ISOLATED LARGE RESIDUAL. (ICTYP = 6)
C
      ELSE IF (ABS(SI).GT.1.5*SBIG(II)
& .AND.(ABS(SJ).LT.SBIG(JJ).AND.JJ.NE.0)) THEN
      ICTYP(JT) = 6
      ICT = 6
      ITST(II) = 1
      ITST(JJ) = 1
      ELSE
      ITST(II) = 1
      ICT = 0
      ENDIF
      IF((ICT.EQ.1).OR.(ICT.EQ.2)) THEN
      GO TO 100
      ELSE
      GO TO 300
      ENDIF
      END
C*****
      SUBROUTINE FNEXT(S,SBIG,ITST,KRES,ISL,ISU,II,JJ,SI,SJ)
C
C FIND THE (NEXT) LARGEST TEMPERATURE RESIDUAL, S(II)/SBIG(II)
C AND LARGEST ADJACENT TEMPERATURE RESIDUAL, S(JJ)/SBIG(JJ)
C AMONG LAYERS THAT HAVE NOT ALREADY BEEN TESTED.
C A PRIORITY SYSTEM IS USED FOR THE CORRECTIONS OR
C SUGGESTED CORRECTIONS. THE LAYER WITH THE LARGEST
C RESIDUAL IS CHOSEN, BUT MID, THIN LAYERS ARE GIVEN TOP
C PRIORITY, THEN TOP OR BOTTOM LAYERS, AND OTHER CASES
C ARE GIVEN LAST PRIORITY. THE NEIGHBORS ARE ALSO SEARCHED
C WITH PRIORITY. LARGEST THIN NEIGHBOR FIRST, THEN OTHERS.
C (NEW VERSION 6/20/88)
C
      DIMENSION S(1), SBIG(1), ITST(1), ISL(1), ISU(1)
      II = 0
      JJ = 0
      IIMID = 0
      IIBT = 0
      IITHK = 0
      SMID = 0.
      SBT = 0.
      STHK = 0.

```



```

SI = 0.
SJ = 0.
RATO = 0.
DO 200 I=1,KRES
  IF(ITST(I).EQ.0) THEN
    RAT = ABS(S(I)/SBIG(I))
    IDIF = ISU(I) - ISL(I)
    IF(RAT.GT.RATO.AND.IDIF.LE.2.AND.

```

```

&      I.NE.1.AND.I.NE.KRES) THEN
      IIMID = I
      SMID = S(I)
      RATO = RAT
    ENDIF
  ENDIF

```

```

200 CONTINUE

```

```

RATO = 0.

```

```

DO 210 I=1,KRES

```

```

  IF(ITST(I).EQ.0) THEN
    RAT = ABS(S(I)/SBIG(I))
    IF(RAT.GT.RATO.AND.

```

```

&      (I.EQ.1.OR.I.EQ.KRES)) THEN
      IIBT = I
      SBT = S(I)
      RATO = RAT
    ENDIF
  ENDIF

```

```

210 CONTINUE

```

```

RATO = 0.

```

```

DO 220 I=1,KRES

```

```

  IF(ITST(I).EQ.0) THEN
    RAT = ABS(S(I)/SBIG(I))
    IF(RAT.GT.RATO) THEN

```

```

      IITHK = I
      STHK = S(I)
      RATO = RAT
    ENDIF
  ENDIF

```

```

220 CONTINUE

```

C  
C  
C  
C

```

CHOOSE II AMONG POSSIBLE VALUES IN THE PRIORITY:
  1. IIMID,  2. IIBT,  3. IITHK

```

```

IF(IIMID.NE.0.AND.ABS(SMID).GT.SBIG(IIMID)) THEN
  II = IIMID
ELSE IF(IIBT.NE.0.AND.ABS(SBT).GT.SBIG(IIBT)) THEN
  II = IIBT
ELSE
  II = IITHK
ENDIF

```

```

SI = S(II)

```

```

IF(II.EQ.0) GO TO 390

```

```

IF(KRES.EQ.1) GO TO 390

```

C  
C  
C  
C  
C  
C  
C

```

NOW, FIND LARGEST NEIGHBOR.

```

```

IF JJ REMAINS SET = 0, THEN NO UNCHECKED NEIGHBORS
EXIST.

```

```

CHECK FOR BOTTOM.

```

```
IF(II.EQ.1) THEN
  IF(ITST(2).EQ.0) THEN
    JJ = 2
  ELSE
    JJ = 0
  ENDIF
```

C  
C  
C

```
CHECK FOR TOP.
```

```
ELSE IF(II.EQ.KRES) THEN
  IF(ITST(KRES-1).EQ.0) THEN
    JJ = KRES - 1
  ELSE
    JJ = 0
  ENDIF
```

C  
C  
C

```
LAYERS ARE IN MIDDLE. SEE IF ALREADY TESTED.
```

```
ELSE
  IIM = II - 1
  IIP = II + 1
  IF(ITST(IIM).EQ.1.AND.ITST(IIP).EQ.0) THEN
    JJ = IIP
  ELSE IF(ITST(IIM).EQ.0.AND.ITST(IIP).EQ.1) THEN
    JJ = IIM
  GIVE PRIORITY TO THIN NEIGHBOR.
  ELSE IF(ITST(IIM).EQ.0.AND.ITST(IIP).EQ.0) THEN
    SM = S(IIM)/SBIG(IIM)
    SP = S(IIP)/SBIG(IIP)
    IDIFM = ISU(IIM) - ISL(IIM)
    IDIFP = ISU(IIP) - ISL(IIP)
    IF(IDIFM.LE.2.AND.IDIFP.GT.2) THEN
      JJ = IIM
    ELSEIF(IDIFP.LE.2.AND.IDIFM.GE.2) THEN
      JJ = IIP
    ELSE
      IF(ABS(SM).GT.ABS(SP)) JJ = IIM
      IF(ABS(SP).GT.ABS(SM)) JJ = IIP
    ENDIF
  ENDIF
```

C

```
ENDIF
ENDIF
SJ = S(JJ)
390 CONTINUE
RETURN
END
```

C\*\*\*\*\*

```
SUBROUTINE TEST(ITST,KRES,IT)
DIMENSION ITST(1)
IT = 0
DO 10 I=1,KRES
  IT = IT + ITST(I)
10 CONTINUE
RETURN
END
```

C\*\*\*\*\*

SUBROUTINE RESID(Z,T,MAN,S,SBIG,BSUM,ISL,ISU,KMAX)  
C\$\$\$ SUBPROGRAM DOCUMENTATION BLOCK

C  
C SUBPROGRAM: RESID CALCULATE HYDROSTATIC RESIDUALS  
C PRGMMR: W. COLLINS ORG: W/NMC22 DATE: 88-09-14  
C

C ABSTRACT: CALCULATE HYDROSTATIC RESIDUALS FOR MANDATORY  
C LAYERS. ACCOUNT FOR MISSING.  
C

C PROGRAM HISTORY LOG:  
C 88009-14 ORIGINAL W. COLLINS  
C

C USAGE: CALL RESID(Z, T, MAN, S, SBIG, BSUM, ISL,  
C ISU, KMAX)  
C

C INPUT ARGUMENT LIST:  
C Z - HEIGHT PROFILE  
C T - TEMPERATURE PROFILE  
C MAN - NUMBER OF MANDATORY LEVELS  
C SBIG - ADMISSIBLE RESIDUALS  
C OUTPUT ARGUMENT LIST:  
C S - HYDROSTATIC RESIDUALS  
C BSUM - B FOR OUTPUT LAYERS  
C ISL - INDICES FOR LOWER LAYER LIMITS  
C ISU - INDICES FOR UPPER LAYER LIMITS  
C KMAX - NUMBER OF OUTPUT LAYERS  
C

C ATTRIBUTES:  
C LANGUAGE: VS FORTRAN  
C MACHINE: NAS  
C

C\$\$\$

COMMON /CONSTS/ R, G, TO, A(54), B(54), SS(54)  
DIMENSION Z(1), T(1), S(1), SBIG(1), ISL(1), ISU(1),  
& BSUM(1)  
DATA ZMSG/99999./, TMSG/9999.9/, ZMAX/90000./, TMAX/9000./  
RES(Z1,Z2,T1,T2,A0,B0) = Z2 - Z1 - A0 - B0 \* (T1+T2)  
MANM = MAN - 1

C  
C COMPUTE RESIDUALS.  
C

DO 10 K=1,54  
S(K) = 0.  
SBIG(K) = 0.  
BSUM(K) = 1.E-6  
ISL(K) = 0  
ISU(K) = 0  
10 CONTINUE  
IK = 1  
KMAX = IK - 1  
DO 50 K=1,MANM  
IF(Z(K).GT.ZMAX.OR.T(K).GT.TMAX) GO TO 50

C  
C KLOW IS LOWER LEVEL WITHOUT MISSING DATA.  
C

KLOW = K  
KP = K + 1  
DO 20 KK=KP,MAN  
KHIGH = KK  
IF(Z(KK).LT.ZMAX.AND.T(KK).LT.TMAX) GO TO 25

20 CONTINUE  
GO TO 50

C  
C  
C

KHIGH IS UPPER LEVEL WITHOUT MISSING DATA.

25 ASUM = 0.  
SSUM = 0.  
KHIGHM = KHIGH - 1  
ISL(IK) = KLOW  
ISU(IK) = KHIGH  
DO 30 KK=KLOW,KHIGHM  
ASUM = ASUM + A(KK)  
BSUM(IK) = BSUM(IK) + B(KK)  
SSUM = SSUM + SS(KK)\*\*2

30 CONTINUE  
SBIG(IK) = SQRT(SSUM)  
S(IK)=RES(Z(KLOW), Z(KHIGH), T(KLOW), T(KHIGH),  
& ASUM, BSUM(IK))  
IK = IK + 1

C  
C  
C

KMAX IS NUMBER OF LEVELS OF RESIDUALS.

KMAX = IK - 1  
50 CONTINUE  
RETURN  
END